######                 ######
###  QMAKE Configuration  ###
######                 ######

TEMPLATE	= app
TARGET	= recordstation
BUILD_PATH	= ./build
CONFIG(debug, debug|release) {
	# Debug
	CONFIG	+= console
	TARGET	= $$join(TARGET,,,_debug)
	unix:OBJECTS_DIR	= $${BUILD_PATH}/debug/.obj/unix
	win32:OBJECTS_DIR	= $${BUILD_PATH}/debug/.obj/win32
	mac:OBJECTS_DIR	= $${BUILD_PATH}/debug/.obj/mac
	win32:UI_DIR	= ./
	unix:UI_DIR	= ./
	MOC_DIR	= $${BUILD_PATH}/debug/.moc
	RCC_DIR	= $${BUILD_PATH}/debug/.rcc
} else {
	# Release
	TARGET	= $$join(TARGET,,,_release)
	unix:OBJECTS_DIR	= $${BUILD_PATH}/release/.obj/unix
	win32:OBJECTS_DIR	= $${BUILD_PATH}/release/.obj/win32
	mac:OBJECTS_DIR	= $${BUILD_PATH}/release/.obj/mac
	win32:UI_DIR	= ./
	unix:UI_DIR	= ./
	MOC_DIR	= $${BUILD_PATH}/release/.moc
	RCC_DIR	= $${BUILD_PATH}/release/.rcc
}
CONFIG( debug, debug|release ) {
	PACKAGE_VERSION	= $${PACKAGE_VERSION}_debug
}

# define variable for source code
DEFINES	*= "PACKAGE_NAME=\"\\\"$${QMAKE_TARGET_PRODUCT}\\\"\"" \
	"PACKAGE_VERSION=\"\\\"$${PACKAGE_VERSION}\\\"\"" \
	"PACKAGE_DOMAIN=\"\\\"$${PACKAGE_DOMAIN}\\\"\"" \
	"PACKAGE_COPYRIGHTS=\"\\\"$${QMAKE_TARGET_COPYRIGHT}\\\"\"" \
	"PACKAGE_COMPANY=\"\\\"$${QMAKE_TARGET_COMPANY}\\\"\""

DEPENDPATH	= . \
	Forms

INCLUDEPATH	= .

# Input
HEADERS	+= Classes/MainWindow/MainWindow.h \
	Classes/HMISuperviseur/HMISuperviseur.h \
	Classes/StunSlider/StunSlider.h \
	Classes/Record/AudioRec.h
SOURCES	= main.cpp \
	Classes/MainWindow/MainWindow.cpp \
	Classes/HMISuperviseur/HMISuperviseur.cpp \
	Classes/StunSlider/StunSlider.cpp \
	Classes/Record/AudioRec.cpp

FORMS	+= Forms/MainWindow.ui
RESOURCES	= Forms/All_Resources.qrc

RC_FILE	= Forms/mainicon.rc

QT	= core gui

